Row Reduction, Partial Pivoting, and PLU Decomposition in Maple

The "LinearAlgebra" package contains all of the matrix operation functions that we need. Use "<" and ">" to begin and end rows/columns. A comma "," moves down a row and a pipe "|" moves over a column.

"RowOperation" will perform row operations on a matrix.

Type I: RowOperation(A,[i,j]); swaps matrix A's rows i and j.
Type II: RowOperation(A,i,c); scales matrix A's row i by c.
Type III: RowOperation(A,[i,j],c); adds c times matrix A's row j to its row i.

By using "inplace=true", the RowOperation command will overwrite A with the result of performing the desired operation.

```
> restart;
  with(LinearAlgebra):
```

Two ways to punch in a matrix (by columns or by rows)...

```
> A := <<1,2,1>|<2,4,2>|<1,1,0>|<3,6,1>|<0,1,1>>;
```

$$A := \begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 2 & 4 & 1 & 6 & 1 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad (1)$$

```
> A := <<1|2|1|3|0>,<2|4|1|6|1>,<1|2|0|1|1>>;
```

$$A := \begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 2 & 4 & 1 & 6 & 1 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad (2)$$

Swapping rows 1 and 2, scaling row 2 by 5, and adding 3 times row 2 to row 1...

```
> RowOperation(A,[1,2]);
```

$$\begin{bmatrix} 2 & 4 & 1 & 6 & 1 \\ 1 & 2 & 1 & 3 & 0 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad (3)$$

```
> RowOperation(A,2,5);
```

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 10 & 20 & 5 & 30 & 5 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad (4)$$

```
> RowOperation(A,[1,2],3);
```

$$\begin{bmatrix} 7 & 14 & 4 & 21 & 3 \\ 2 & 4 & 1 & 6 & 1 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad (5)$$

Row reducing as in class...

```
> A;
```

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 2 & 4 & 1 & 6 & 1 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad (6)$$

```
> RowOperation(A,[2,1],-2,inplace=true);
```

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 0 & 0 & -1 & 0 & 1 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad (7)$$

```
> RowOperation(A,[3,1],-1,inplace=true);
```

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 0 & 0 & -1 & 0 & 1 \\ 0 & 0 & -1 & -2 & 1 \end{bmatrix} \quad (8)$$

```
> RowOperation(A,[3,2],-1,inplace=true);
```

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 0 & 0 & -1 & 0 & 1 \\ 0 & 0 & 0 & -2 & 0 \end{bmatrix} \quad (9)$$

Forward pass...done!

```
> RowOperation(A,3,-1/2,inplace=true);
```

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 0 & 0 & -1 & 0 & 1 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix} \quad (10)$$

```
> RowOperation(A,[1,3],-3,inplace=true);
```

$$\begin{bmatrix} 1 & 2 & 1 & 0 & 0 \\ 0 & 0 & -1 & 0 & 1 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix} \quad (11)$$

```
> RowOperation(A,2,-1,inplace=true);
```

(12)

$$\begin{bmatrix} 1 & 2 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix} \quad (12)$$

```
> RowOperation(A,[1,2],-1,inplace=true);
```

$$\begin{bmatrix} 1 & 2 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix} \quad (13)$$

Backward pass...done!

This is the RREF of A.

```
> A := <<1,2,1>|<2,4,2>|<1,1,0>|<3,6,1>|<0,1,1>>;
```

$$A := \begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 2 & 4 & 1 & 6 & 1 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad (14)$$

```
> ReducedRowEchelonForm(A);
```

$$\begin{bmatrix} 1 & 2 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix} \quad (15)$$

Same thing with elementary matrices...

Recall that an elementary matrix is simply the identity matrix after a single operation is performed on it.

```
> E[1] := RowOperation(IdentityMatrix(3),[2,1],-2);
  E[2] := RowOperation(IdentityMatrix(3),[3,1],-1);
  E[3] := RowOperation(IdentityMatrix(3),[3,2],-1);
  E[4] := RowOperation(IdentityMatrix(3),3,-1/2);
  E[5] := RowOperation(IdentityMatrix(3),[1,3],-3);
  E[6] := RowOperation(IdentityMatrix(3),2,-1);
  E[7] := RowOperation(IdentityMatrix(3),[1,2],-1);
```

$$E_1 := \begin{bmatrix} 1 & 0 & 0 \\ -2 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$E_2 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ -1 & 0 & 1 \end{bmatrix}$$

$$E_3 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & -1 & 1 \end{bmatrix}$$

$$E_4 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & -\frac{1}{2} \end{bmatrix}$$

$$E_5 := \begin{bmatrix} 1 & 0 & -3 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$E_6 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$E_7 := \begin{bmatrix} 1 & -1 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \qquad \textbf{(16)}$$

```
> E[1].A;
  E[2].E[1].A;
  E[3].E[2].E[1].A;
  E[4].E[3].E[2].E[1].A;
  E[5].E[4].E[3].E[2].E[1].A;
  E[6].E[5].E[4].E[3].E[2].E[1].A;
  E[7].E[6].E[5].E[4].E[3].E[2].E[1].A;
```

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 0 & 0 & -1 & 0 & 1 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 0 & 0 & -1 & 0 & 1 \\ 0 & 0 & -1 & -2 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 0 & 0 & -1 & 0 & 1 \\ 0 & 0 & 0 & -2 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 0 & 0 & -1 & 0 & 1 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 1 & 0 & 0 \\ 0 & 0 & -1 & 0 & 1 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix} \quad \textbf{(17)}$$

In other words, if $P = E_7.E_6.E_5.E_4.E_3.E_2.E_1$, then PA will be the RREF of A.

```
> P := 'E[7].E[6].E[5].E[4].E[3].E[2].E[1]';
```

$$P := E_7 \cdot E_6 \cdot E_5 \cdot E_4 \cdot E_3 \cdot E_2 \cdot E_1 \quad \textbf{(18)}$$

```
> P.A;
```

$$\begin{bmatrix} 1 & 2 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix} \quad \textbf{(19)}$$

Also, operations 1-3 correspond to the forward pass and 4-7 correspond to the backward pass. So if $B = E_3.E_2.E_1$, then $BA = U$ is the REF resulting from the forward pass. This also means that $A = B^{-1}U$ $= LU$.

```
> U := E[3].E[2].E[1].A;
  L := E[1]^(-1).E[2]^(-1).E[3]^(-1);
```

$$U := \begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 0 & 0 & -1 & 0 & 1 \\ 0 & 0 & 0 & -2 & 0 \end{bmatrix}$$

$$L := \begin{bmatrix} 1 & 0 & 0 \\ 2 & 1 & 0 \\ 1 & 1 & 1 \end{bmatrix} \quad \textbf{(20)}$$

```
> A=L.U;
```

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 2 & 4 & 1 & 6 & 1 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} = \begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 2 & 4 & 1 & 6 & 1 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad \textbf{(21)}$$

Let's find the PLU decomposition of the same matrix A if I demand that we do partial pivoting!

```
> A;
```

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 2 & 4 & 1 & 6 & 1 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad (22)$$

```
> RowOperation(A,[1,2],inplace=true);
```

$$\begin{bmatrix} 2 & 4 & 1 & 6 & 1 \\ 1 & 2 & 1 & 3 & 0 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad (23)$$

```
> RowOperation(A,[2,1],-1/2,inplace=true);
```

$$\begin{bmatrix} 2 & 4 & 1 & 6 & 1 \\ 0 & 0 & \frac{1}{2} & 0 & -\frac{1}{2} \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad (24)$$

```
> RowOperation(A,[3,1],-1/2,inplace=true);
```

$$\begin{bmatrix} 2 & 4 & 1 & 6 & 1 \\ 0 & 0 & \frac{1}{2} & 0 & -\frac{1}{2} \\ 0 & 0 & -\frac{1}{2} & -2 & \frac{1}{2} \end{bmatrix} \quad (25)$$

```
> RowOperation(A,[3,2],1,inplace=true);
```

$$\begin{bmatrix} 2 & 4 & 1 & 6 & 1 \\ 0 & 0 & \frac{1}{2} & 0 & -\frac{1}{2} \\ 0 & 0 & 0 & -2 & 0 \end{bmatrix} \quad (26)$$

The forward pass is complete. This is our "U". If we had been doing our book keeping we'd also know our "L". I'll just type it in...

```
> U := A;
  L := <<1,1/2,1/2>|<0,1,-1>|<0,0,1>>;
```

$$U := \begin{bmatrix} 2 & 4 & 1 & 6 & 1 \\ 0 & 0 & \frac{1}{2} & 0 & -\frac{1}{2} \\ 0 & 0 & 0 & -2 & 0 \end{bmatrix}$$

(27)

$$L := \begin{bmatrix} 1 & 0 & 0 \\ \frac{1}{2} & 1 & 0 \\ \frac{1}{2} & -1 & 1 \end{bmatrix} \quad (27)$$

The "P" part of the PLU decomposition comes from the row swaps we did. In this case there was just one. "P" is the accumulated row swaps all "undone". In our case, we just need to undo a single swap of rows 1 and 2 (which again is just a swap of rows 1 and 2)...

```
> P := RowOperation(IdentityMatrix(3),[1,2]);
```

$$P := \begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (28)$$

```
> A := P.L.U;
```

$$A := \begin{bmatrix} 1 & 2 & 1 & 3 & 0 \\ 2 & 4 & 1 & 6 & 1 \\ 1 & 2 & 0 & 1 & 1 \end{bmatrix} \quad (29)$$

Let's solve the system Ax=b where b is defined below...

```
> b := <1,2,3>;
```

$$b := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix} \quad (30)$$

Solve...

```
> 'P.L.U'.<<x[1],x[2],x[3],x[4],x[5]>> = b;
```

$$P \;.\; L \;.\; U \;.\; \begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \\ x_5 \end{bmatrix} = \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix} \quad (31)$$

```
> 'L.U'.<<x[1],x[2],x[3],x[4],x[5]>> = P^(-1).b;
```

$$L\ .\ U\ .\ \begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \\ x_5 \end{bmatrix} = \begin{bmatrix} 2 \\ 1 \\ 3 \end{bmatrix} \quad \textbf{(32)}$$

First, we solve $L\,y = P^{-1}b$ (where $y = U\,x$ will be solved later)...

```
> L.<<y[1],y[2],y[3]>> = P^(-1).b;
```

$$\begin{bmatrix} y_1 \\ \frac{1}{2}\,y_1 + y_2 \\ \frac{1}{2}\,y_1 - y_2 + y_3 \end{bmatrix} = \begin{bmatrix} 2 \\ 1 \\ 3 \end{bmatrix} \quad \textbf{(33)}$$

```
> y[1]:=2;
  y[2]:=1-1/2*y[1];
  y[3]:=3-1/2*y[1]+y[2];
```

$$y_1 := 2$$
$$y_2 := 0$$
$$y_3 := 2 \quad \textbf{(34)}$$

Finally, we need to solve $U\,x = y$...

```
> U.<<x[1],x[2],x[3],x[4],x[5]>> = <<y[1],y[2],y[3]>>;
```

$$\begin{bmatrix} 2\,x_1 + 4\,x_2 + x_3 + 6\,x_4 + x_5 \\ \frac{1}{2}\,x_3 - \frac{1}{2}\,x_5 \\ -2\,x_4 \end{bmatrix} = \begin{bmatrix} 2 \\ 0 \\ 2 \end{bmatrix} \quad \textbf{(35)}$$

```
> x[5] := s;
  x[4] := -1/2*(2);
  x[3] := 2*(1/2*x[5]);
  x[2] := t;
  x[1] := 1/2*(2-4*x[2]-x[3]-6*x[4]-x[5]);
```

$$x_5 := s$$
$$x_4 := -1$$
$$x_3 := s$$
$$x_2 := t$$

$$x_1 := 4 - 2\,t - s \tag{36}$$

```
> Solution := <<x[1],x[2],x[3],x[4],x[5]>>;
```

$$Solution := \begin{bmatrix} 4 - 2\,t - s \\ t \\ s \\ -1 \\ s \end{bmatrix} \tag{37}$$

Of course this is all overkill. It is much easier just to directly solve...

```
> x := 'x';
  <A|b>;
```

$$x := x$$

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 0 & 1 \\ 2 & 4 & 1 & 6 & 1 & 2 \\ 1 & 2 & 0 & 1 & 1 & 3 \end{bmatrix} \tag{38}$$

```
> ReducedRowEchelonForm(<A|b>);
```

$$\begin{bmatrix} 1 & 2 & 0 & 0 & 1 & 4 \\ 0 & 0 & 1 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 & 0 & -1 \end{bmatrix} \tag{39}$$

```
> x[5] := s;
  x[4] := -1;
  x[3] := x[5];
  x[2] := t;
  x[1] := 4-2*x[2]-x[5];
```

$$x_5 := s$$

$$x_4 := -1$$

$$x_3 := s$$

$$x_2 := t$$

$$x_1 := 4 - 2\,t - s \tag{40}$$

```
> Solution := <<x[1],x[2],x[3],x[4],x[5]>>;
```

$$Solution := \begin{bmatrix} 4 - 2\,t - s \\ t \\ s \\ -1 \\ s \end{bmatrix} \tag{41}$$

Example 2: PLU Decomposition (with partial pivoting) of a 3x3 matrix...

I will build L^{-1} along the way (as ?explained? in class)...

```
> A := <<1,2,4>|<-3,2,1>|<1,0,1>>;
  A^(-1);
```

$$A := \begin{bmatrix} 1 & -3 & 1 \\ 2 & 2 & 0 \\ 4 & 1 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & -1 \\ -1 & -\frac{3}{2} & 1 \\ -3 & -\frac{13}{2} & 4 \end{bmatrix} \quad (42)$$

The last 3 columns are the identity matrix, so as we operate on B, we can see the effect of the row ops on I.

```
> B := <A|IdentityMatrix(3)>;
```

$$B := \begin{bmatrix} 1 & -3 & 1 & 1 & 0 & 0 \\ 2 & 2 & 0 & 0 & 1 & 0 \\ 4 & 1 & 1 & 0 & 0 & 1 \end{bmatrix} \quad (43)$$

```
> RowOperation(B,[1,3],inplace=true);
```

$$\begin{bmatrix} 4 & 1 & 1 & 0 & 0 & 1 \\ 2 & 2 & 0 & 0 & 1 & 0 \\ 1 & -3 & 1 & 1 & 0 & 0 \end{bmatrix} \quad (44)$$

```
> RowOperation(B,[2,1],-1/2,inplace=true);
```

$$\begin{bmatrix} 4 & 1 & 1 & 0 & 0 & 1 \\ 0 & \frac{3}{2} & -\frac{1}{2} & 0 & 1 & -\frac{1}{2} \\ 1 & -3 & 1 & 1 & 0 & 0 \end{bmatrix} \quad (45)$$

```
> RowOperation(B,[3,1],-1/4,inplace=true);
```

$$\begin{bmatrix} 4 & 1 & 1 & 0 & 0 & 1 \\ 0 & \frac{3}{2} & -\frac{1}{2} & 0 & 1 & -\frac{1}{2} \\ 0 & -\frac{13}{4} & \frac{3}{4} & 1 & 0 & -\frac{1}{4} \end{bmatrix} \quad (46)$$

```
> RowOperation(B,[2,3],inplace=true);
```

$$\begin{bmatrix} 4 & 1 & 1 & 0 & 0 & 1 \\ 0 & -\frac{13}{4} & \frac{3}{4} & 1 & 0 & -\frac{1}{4} \\ 0 & \frac{3}{2} & -\frac{1}{2} & 0 & 1 & -\frac{1}{2} \end{bmatrix} \tag{47}$$

```
> RowOperation(B,[3,2],6/13,inplace=true);
```

$$\begin{bmatrix} 4 & 1 & 1 & 0 & 0 & 1 \\ 0 & -\frac{13}{4} & \frac{3}{4} & 1 & 0 & -\frac{1}{4} \\ 0 & 0 & -\frac{2}{13} & \frac{6}{13} & 1 & -\frac{8}{13} \end{bmatrix} \tag{48}$$

```
> U:=B[1..3,1..3];
```

$$U := \begin{bmatrix} 4 & 1 & 1 \\ 0 & -\frac{13}{4} & \frac{3}{4} \\ 0 & 0 & -\frac{2}{13} \end{bmatrix} \tag{49}$$

```
> L:=RowOperation(RowOperation(B[1..3,4..6]^(-1),[1,3]),[2,3]);
```

$$L := \begin{bmatrix} 1 & 0 & 0 \\ \frac{1}{4} & 1 & 0 \\ \frac{1}{2} & -\frac{6}{13} & 1 \end{bmatrix} \tag{50}$$

```
> P:=RowOperation(RowOperation(IdentityMatrix(3),[2,3]),[1,3]);
```

$$P := \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 1 & 0 & 0 \end{bmatrix} \tag{51}$$

```
> A=P.L.U;
```

$$\begin{bmatrix} 1 & -3 & 1 \\ 2 & 2 & 0 \\ 4 & 1 & 1 \end{bmatrix} = \begin{bmatrix} 1 & -3 & 1 \\ 2 & 2 & 0 \\ 4 & 1 & 1 \end{bmatrix} \tag{52}$$

Solving the system $A\,x = b$ with b defined below and using our PLU decomposition...

This time instead of carefully working through the forward and back substitutions I will just use the inverses of our lower and upper triangular matrices (we couldn't do this before - with U anyway - because A wasn't invertible - we had free variables).

```
> b := <<-2,6,9>>;
```

$$b := \begin{bmatrix} -2 \\ 6 \\ 9 \end{bmatrix} \quad (53)$$

```
> 'P.L.U'.x=b;
```

$$P \,.\, L \,.\, U \,.\, x = \begin{bmatrix} -2 \\ 6 \\ 9 \end{bmatrix} \quad (54)$$

```
> 'L.U'.x=P^(-1).b;
```

$$L \,.\, U \,.\, x = \begin{bmatrix} 9 \\ -2 \\ 6 \end{bmatrix} \quad (55)$$

```
> 'U'.x=L^(-1).P^(-1).b;
```

$$U \,.\, x = \begin{bmatrix} 9 \\ -\frac{17}{4} \\ -\frac{6}{13} \end{bmatrix} \quad (56)$$

```
> x = U^(-1).L^(-1).P^(-1).b;
```

$$x = \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix} \quad (57)$$

Directly...

```
> ReducedRowEchelonForm(<A|b>);
```

$$\begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & 2 \\ 0 & 0 & 1 & 3 \end{bmatrix} \quad (58)$$

Maple will find PLU decompositions for you (but without partial pivoting)...

Note: I'm not saying there isn't a command to do PLU decomposition with partial pivoting for you, but I'm not going to find it for you (nor is it worth your time trying to hunt such a procedure down).

```
> ? LUDecomposition
> P, L, U := LUDecomposition(A);
```

$$P, L, U := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 2 & 1 & 0 \\ 4 & \frac{13}{8} & 1 \end{bmatrix}, \begin{bmatrix} 1 & -3 & 1 \\ 0 & 8 & -2 \\ 0 & 0 & \frac{1}{4} \end{bmatrix} \tag{59}$$